EViews LogL オブジェクトの利用方法

はじめに

EViews の Equation オブジェクトによる最小二乗法推定と誤差項に関する診断 (系列相関や不均一分散) に慣れてきたら、そろそろ最尤法推定にチャレンジしてみましょう。ここでは EViews で最尤法推定を行うための LogL オブジェクトの操作方法をご紹介します。ただし、読者として最尤法の基礎知識を持っている方を対象に説明します。

EViews の書式

最初に EViews のサンプルファイル demo.wf1 を画面に開きます。[1]シリーズオブジェクトとしてマネーサプライ m1 と GDP のシリーズ gdp があります。



この変数を使ってまずは最小二乗法 (OLS) で次式を推定します。

$$\text{m1} = \alpha + \beta \times \text{gdp} + u \tag{1}$$

Equation オブジェクト名は myols とします。myols で View/Representations と操作します。EViews は中段の Equation Estimation のところに次のように表示します。

```
m1=c(1)+c(2)*gdp
```

c(1) が α,c(2) が β に対応しています。ここまでは EViews の書式の確認です。c(1) と c(2) はワークファイル画面の β というアイコン (オブジェクト名は c) の中に上から順番に入っています。アイコンをダブルクリックして中身を確認してみましょう。次に説明する LogL(ログエル) オブジェクトではこの書式を利用しますので、正しく理解してください。

[1] Program Files\EViews7\Example Files\EV Manual Data\Chapter 2 にあります。

最尤法による推定

ここから LogL(ログエル) を利用した最尤法の話にはいります。残差 ϵ を次のように定義してみます。α と β は式-1 のそれとは別物とお考えください。

$$\epsilon = \mathrm{m1} - \alpha - \beta\,\mathrm{gdp} \tag{2}$$

ϵ の分散を仮に σ^2 とします。つまり、$\epsilon \sim N\left(0, \sigma^2\right)$ と仮定します。分散 σ^2 を使って標準化した残差 e は次のように書くことができます。

$$e = \frac{\epsilon}{\sqrt{\sigma^2}} \tag{3}$$

$e \sim N\,(0, 1)$ です。

さて、EViews の操作に入ります。最初に mylogl という名前の LogL オブジェクトを作成し、次のように記述します。ここでは ϵ を res、ϵ の分散 var と書くことにします。つまり、

```
@logl name
res=m1-c(1)-c(2)*gdp
var=c(3)
name=log(@dnorm(res/@sqrt(var)))-log(var)/2...(A)
```

4 行目の式を A とします。LoL オブジェクトで Estimate ボタンをクリックすれば推定値を得ます。しかし、実際に推定を行ってみるとエラーが出ます。エラーメッセージは「Missing value...」で、あたかも欠損値があるような雰囲気です。OLS とは異なり、最尤法では欠損値は許されません。データを確認すると分かるのですが、変数 m1 と gdp に欠損値はありません。データは全て揃っています。では、何がいけなかったのでしょうか。初めて最尤法にチャレンジする方が躓くのがこの辺りです。

解決方法は簡単です。OLS 推定した myols の推定結果を見てください。下の表の所で S.E of regression が 23.61657 となっています。回帰の標準誤差です。これを二乗したものがほぼ最尤法推定した時の残差の分散に等しいと考えることができるでしょう。従って、アイコン β をダブルクリックして開き、c(3) のところに 23 を二乗した 529 と入力します。そして、再度推定してみましょう。今度はエラーが出ないと思います。

初期値 最尤法推定では初期値の設定がとても重要であると同時に、実用上、難しいところです。EViews で c(1),c(2),c(3)... というものはパラメータの値または**初期値**を示します。実体はワークファイルウィンドウの β というアイコン (名前は係数を示す c) の中にあります。この係数オブジェクトの分散の初期値に相当するセルに注目してください。これが実際の残差の分散にある程度、近しくなければエラーになったり、本来とは異なる値を推定してしまいます。ここを上手に処理する方法 (テクニック) は後半でご説明します。

コマンドの説明

上に記述した 4 行のコマンド行で利用したコマンドの意味を確認しておきます。

`@logl`: 尤度関数を宣言するコマンドです。スペースを空けてその後ろに自分の好きな名前の尤度関数名を記述します。ここでは name としました。

`@dnorm`(変数名): 標準正規分布 $N(0,1)$ の確率密度を求める関数です。

`@sqrt`(変数名): 平方根を求めます。

具体的に言うと、各行では次のような処理を行っています。

1 行目:name という名前の対数尤度関数を宣言します。
2 行目:密度の対象となる変数 res を求めます。
3 行目:残差の分散を c(3) とします。均一分散とします。
4 行目:対尤度関数 name を具体的に定義します。

最尤法推定はこの 1-4 行目の繰り返し計算を行います。複雑そうに見えますね。しかし、がんばって 4 行目の計算式について理解しましょう。ここをある程度理解しないと、LogL オブジェクトを使えるようになりません。残差 ϵ が最小二乗法の場合と同じく、$N\left(0,\sigma^2\right)$ に従うとすると、その確率密度関数 $f(\epsilon)$ は次のうようなものです。ご存知、ガウス分布関数です。

$$f(\epsilon)=\frac{1}{\sqrt{2\pi\sigma^2}}\exp\left[-\frac{\epsilon^2}{2\sigma^2}\right] \tag{4}$$

これはベル型曲線の縦軸の値 (密度) を算出します。横軸の値は ϵ です。この式の対数をとった対数尤度を EViews のコマンドで記述しますと、次のようになります。これを B とします。

```
name=log(@sqrt(2*4*@atan(1)*var)^(-1))-res^2/(2*var)...(B)
```
[2]

この (B) は (A) と同じことなのです。式-3 を使って式-4 を少し変形してみましょう。つまり、ϵ の代わりに標準化した e を利用して、

$$\begin{aligned} f(e) &= \frac{1}{\sqrt{2\pi\sigma^2}}\exp\left[-e^2\cdot\sigma^2\cdot\frac{1}{2\sigma^2}\right] \\ &= \frac{1}{\sqrt{2\pi\sigma^2}}\exp\left[-\frac{e^2}{2}\right] \\ &= \left(\frac{1}{\sqrt{2\pi}}\exp\left[-\frac{e^2}{2}\right]\right)\times\left(\frac{1}{\sqrt{\sigma^2}}\right) \end{aligned} \tag{5}$$

この右辺を EViews のコマンドで書くと、次のようになります。

```
@dnorm(res/*@sqrt(var))*(1/@sqrt(var))
```

さらに対数をとります。

```
log(@dnorm(res/@sqrt(var)))+log(1/@sqrt(var))
=log(@dnorm(res/@sqrt(var)))-log(var)/2
```

として、A と同じになりました。結論は A と B 好きな方で書いてください、ということです。マニュアルや書籍では普通、A の形式で記述されています。

[2] EViews では π を $4\times\arctan(1)$ のように計算します。

同時確率密度

最尤法は同時確率密度を最大化するパラメータを推定するものです。いま、n 個のデータがあるとすると尤度 L は次のようになります。

$$L = \frac{1}{\sqrt{2\pi\sigma^2}} \exp\left[-\frac{e_1^2}{2}\right] \cdot \frac{1}{\sqrt{2\pi\sigma^2}} \exp\left[-\frac{e_2^2}{2}\right] \cdots\cdots \frac{1}{\sqrt{2\pi\sigma^2}} \exp\left[-\frac{e_n^2}{2}\right] \tag{6}$$

対数尤度 LL はこの対数ですから、

$$LL = n \times \log\left(\frac{1}{\sqrt{2\pi\sigma^2}}\right) - \frac{\sum_{i=1}^{n} e_i^2}{2} \tag{7}$$

EViews で最尤法を利用する場合、4 行目の右辺に書くべきは、前出の密度関数の式-4 や式-5 であり、式-7 ではないということを覚えてください。

不均一分散について考える

mylogl と myeq の推定結果を比べましょう。ほとんど差はないはずです。線形モデルを OLS 推定した場合、推定値は最尤法のそれと同じになります。値は小数第三位で四捨五入しました。

	係数	t 値
c(1)	70.40	25.1
c(2)	0.58	142.72

OLS による推定

	係数	z 値
c(1)	70.40	8.92
c(2)	0.58	76.83
c(3)	550.88	8.88

最尤法による推定

最尤法による推定ではパラメータの分布を正規分布と仮定しますので z 値を表示します。

次は最尤法推定の特徴が分かる推定を行ってみましょう。myols で resid ボタンをクリックしての残差を不均一分散がある場合について考えます。



残差 Residual の分散はプロットした期間の後半部分で激しく変動しています。実際に myols で View/Residual Diagnostics/Heteroskedasticity Tests... と操作し、Breusch-Pagan-Godfrey Test を実行してみましょう。すると、不均一分散という結論が得られます[3]。OLS 推定の場合、White や HAC オプションを利用して、推定値の標準誤差を不均一分散に対応したものに再推定できます。この時、推定値は変わりません。イメージとして考えるなら、ベル型分布の中心の位置は変わりませんが、ベルの形が変化すると理解します。例として myols でオプションを利用して再推定した結果を次に示します。

	c(1)	t 値	c(2)	t 値
オプションなし	70.40	25.10	0.58	142.72
HAC オプション	70.40	20.92	0.58	52.74
OLS による推定				

もし、私たちが何らの情報をもとに残差の分散を次のように定義できるとしたら、どうしたらいいでしょう?

```
var=c(3)*rs^c(4)
```

LogL では分散も自分で定義できますので、それを使ってみましょう。つまり、新しい LogL オブジェクト mylogl2 に次のように記述します。

```
@logl name
res=m1-c(1)-c(2)*gdp
var=c(3)*rs^c(4)
name=log(@dnorm(res/@sqrt(var)))-log(var)/2
```

c(3) と c(4) にどのような値がはいっているのか、予め β で確認して mylogl2 を推定します。参考までに私の推定結果をここでまとめておきます。

[3] 系列相関も存在します。

	c(1)	z 値	c(2)	z 値	c(3)	z 値	c(4)	z 値
均一分散	70.40	8.92	0.58	76.83	550.88	8.88	–	–
不均一分散	73.17	124.81	0.60	348.66	2.94	1.22	2.87	12.30

最尤法による推定

ここでまEViewsで最尤法推定を行うための基本的な操作方法と関連するEViewsの機能をご説明しました。最尤法の場合には密度関数をどのように書くか、という所がキーになります。ここでご紹介した標準正規分布ばかりなら、たいして難しいことはないのですが、実際には色々な密度関数にチャレンジしなければなりません。つまり、統計学の教科書をしっかり読まないと先へ進めなくなってしまいます。がんばりどころですね。

プログラミング

初期値の設定をもっと効率よくやるためには、EViewsのプログラミングを利用します。実際、EViews 7をインストールすると、一緒にたくさんの、最尤法推定を行うサンプルプログラムがインストールされます[4]。これらの貴重なサンプルプログラムの内容を理解するために、きわめて基本的な最尤推定のためのプログラムをご紹介します。

実際にプログラムを記述する場合、先頭の番号1)などは書かないでください。

```
1)smpl 1952q1 1992q4
2)equation myols2.ls m1 c gdp
3)!v=myols2.@se^2
4)param c(3) !v c(4) 1
5)logl mylogl3
6)mylogl3.append @logl name
7)mylogl3.append res=m1-c(1)-c(2)*gdp
8)mylogl3.append var=c(3)*rs^c(4)
9)mylogl3.append name=log(@dnorm(res/@sqrt(var)))-log(var)/2
10)mylogl3.ml(showopts,m=1000,c=1e-5)
11)show mylogl3.output
```

- 説明

  1行目:念のためサンプル期間を設定します。

  2行目:OLSで推定を行います。

  3行目:回帰の標準誤差を数値変数!vに確保します。

  4行目:パラメータの初期値を設定します。c(4)は経験により1を使います。

  5行目:LogLオブジェクトmylogl3を作成します。

  6-9行目:mylogl3の中身を記述します。

[4] Program Files\EViews7\Example Files\Samle Programs\logl にあります。

10 行目:推定を実行します。[5]

11 行目:推定結果を表示します。

終わりに

EViews ユーザの皆様のために、なるべく分かりやすくと思って入門ガイドのような感じでまとめてみました。少しでもお役に立てば幸いです。ここではご紹介できなかった事柄もまだありますが、ご興味のある方は是非、当社の EViews 講習会「EViews による最尤法入門」をご検討ください。
この資料は Scientific WorkPlace 5.5 という当社の製品を利用して作成しました。こちらも LaTeX を使った論文作成に欠かせないツールです。まずはデモ版をお試しください。
http://www.lightstone.co.jp/latex

株式会社 ライトストーン 2011 年 10 月

[5]カッコ内は Estimate ボタンをクリックした時のダイアログで設定可能な項目のデフォルト値です。